NAG Library Routine Document

G05PXF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05PXF generates a random orthogonal matrix.

2 Specification

```
SUBROUTINE G05PXF(SIDE, INIT, M, N, STATE, A, LDA, IFAIL)
INTEGER           M, N, STATE(lstate), LDA, IFAIL
double precision  A(LDA,N)
CHARACTER*1       SIDE, INIT
```

3 Description

G05PXF pre- or post-multiplies an m by n matrix A by a random orthogonal matrix U, overwriting A. The matrix A may optionally be initialized to the identity matrix before multiplying by U, hence U is returned. U is generated using the method of Stewart (1980). The algorithm can be summarized as follows.

Let $x_1, x_2, \ldots, x_{n-1}$ follow independent multinormal distributions with zero mean and variance $I\sigma^2$ and dimensions $n, n-1, \ldots, 2$; let $H_j = \text{diag}\left(I_{j-1}, H_j^*\right)$, where I_{j-1} is the identity matrix and H_j^* is the Householder transformation that reduces x_j to $r_{jj}e_1$, e_1 being the vector with first element one and the remaining elements zero and r_{jj} being a scalar, and let $D = \text{diag}(\text{sign}(r_{11}), \text{sign}(r_{22}), \ldots, \text{sign}(r_{nn}))$. Then the product $U = DH_1H_2 \ldots H_{n-1}$ is a random orthogonal matrix distributed according to the Haar measure over the set of orthogonal matrices of n. See Theorem 3.3 in Stewart (1980).

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05PXF.

4 References

Stewart G W (1980) The efficient generation of random orthogonal matrices with an application to condition estimates *SIAM J. Numer. Anal.* **17** 403–409

5 Parameters

1: SIDE – CHARACTER*1 *Input*

On entry: indicates whether the matrix A is multiplied on the left or right by the random orthogonal matrix U.

SIDE = 'L'
The matrix A is multiplied on the left, i.e., pre-multiplied.

SIDE = 'R'
The matrix A is multiplied on the right, i.e., post-multiplied.

Constraint: SIDE = 'L' or 'R'.

2: INIT – CHARACTER*1 *Input*

On entry: indicates whether or not A should be initialized to the identity matrix.

INIT = 'I'
A is initialized to the identity matrix.

INIT = 'N'
A is not initialized and the matrix A must be supplied in A.

Constraint: INIT = 'I' or 'N'.

3: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraints:

if SIDE = 'L', M > 1;
otherwise M ≥ 1.

4: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraints:

if SIDE = 'R', N > 1;
otherwise N ≥ 1.

5: STATE($lstate$) – INTEGER array *Communication Array*

Note: $lstate$ = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

6: A(LDA,N) – ***double precision*** array *Input/Output*

On entry: if INIT = 'N', A must contain the matrix A, with the (i,j)th element of A stored in A$((i-1) \times$ LDA $+ j)$.

On exit: the matrix UA when SIDE = 'L' or the matrix AU when SIDE = 'R'.

7: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which G05PXF is called.

Constraint: LDA $\geq$ M.

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, SIDE $\neq$ 'L' or 'R'.

IFAIL $= 2$

On entry, INIT $\neq$ 'I' or 'N'.

IFAIL $= 3$

On entry, M < 1.

IFAIL $= 4$

On entry, N < 1.

IFAIL $= 5$

On entry, STATE vector was not initialized or has been corrupted.

IFAIL $= 7$

On entry, LDA $<$ M.

IFAIL $= 8$

On entry, an orthogonal matrix of dimension 1 has been requested.

7 Accuracy

The maximum error in $U^{\mathrm{T}}U$ should be a modest multiple of ***machine precision*** (see Chapter X02).

8 Further Comments

None.

9 Example

Following initialization of the pseudorandom number generator by a call to G05KFF, a 4 by 4 orthogonal matrix is generated using the INIT $=$ 'I' option and the result printed.

9.1 Program Text

```
*     G05PXF Example Program Text
*     Mark 22 Release. NAG Copyright 2006.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          N, M, LDA
      PARAMETER        (N=4,M=4,LDA=10)
      INTEGER          MSEED, MSTATE
      PARAMETER        (MSEED=1,MSTATE=633)
*     .. Local Scalars ..
      INTEGER          GENID, I, IFAIL, J, LSEED, LSTATE, SUBID
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,N)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*     .. External Subroutines ..
```

```
      EXTERNAL         G05KFF, G05PXF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05PXF Example Program Results'
      WRITE (NOUT,*)
*     Initialize the seed
      SEED(1) = 1762543
*     GENID and SUBID identify the base generator
      GENID = 1
      SUBID = 1
*     Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 40
      END IF
*     Generate the random orthogonal matrix
      IFAIL = 1
      CALL G05PXF('Right','Initialize',M,N,STATE,A,LDA,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 40
      END IF
*     Display the results
      DO 20 I = 1, M
         WRITE (NOUT,99999) (A(I,J),J=1,N)
   20 CONTINUE
*
   40 CONTINUE
*
99999 FORMAT (1X,4F9.3)
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05KPXF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05PXF Example Program Results

     0.176    0.740   -0.307   -0.572
     0.659   -0.578   -0.219   -0.428
     0.668    0.317    0.608    0.290
    -0.297   -0.132    0.699   -0.637
```